

选址问题

姓名：贺正泽 学号：2023302021155

Part1:固定料场，优化运输计划

模型变量

1.决策变量

- x_{ij} : 从料场 j 向工地 i 运送的水泥量

2.其他变量

- x_{i_i} 、 y_{i_i} : 工地的横纵坐标
- x_{j_j} 、 y_{j_j} : 料场的横纵坐标
- d_{ij} : 料场 j 到工地 i 的距离
- a_i : 工地 i 的日用量
- s_j : 料场 j 的最大日运输量

其中 $i = 1, 2, 3, 4, 5, 6; j = 1, 2$ (对应 A, B)

目标函数

最小化：

$$\sum_{i=1}^6 \sum_{j=1}^2 x_{ij} d_{ij}$$

约束条件

1.每个工地的日用量供需平衡

对于 $i = 1, 2, 3, 4, 5, 6$:

$$x_{i1} + x_{i2} = a_i$$

2.每个料场总供应量上限

对于 $j = 1, 2$:

$$\sum_{i=1}^6 x_{ij} \leq s_j$$

3.欧式距离计算

$$d_{ij} = \sqrt{(xi_i - xj_j)^2 + (yi_i - yj_j)^2}$$

4.非负性约束

对于所有的 i, j :

$$x_{ij} \geq 0$$

代码实现

```
SETS:
site /1..6/: xi, yi, ai;
cement /1..2/: xj, yj, supply;
transport(site,cement): x, dist;
ENDSETS

DATA:
xi = 1.25 8.75 0.5 5.75 3 7.25;
yi = 1.25 0.75 4.75 5 6.5 7.75;
ai = 3 5 4 7 6 11;
xj = 5 2;
yj = 1 7;
supply = 20 20;
ENDDATA

! 计算欧几里得距离;
@for(transport(i,j):
    dist(i,j) = @sqrt((xi(i) - xj(j))^2 + (yi(i) - yj(j))^2)
);

! 目标函数: 最小总吨·公里数;
min = @sum(transport(i,j): dist(i,j) * x(i,j));

! 满足每个工地的水泥需求;
@for(site(i):
    @sum(cement(j): x(i,j)) = ai(i)
);

! 每个料场的供应不能超过储存量;
@for(cement(j):
    @sum(site(i): x(i,j)) <= supply(j)
);

END
```

运行结果分析

Model Class:LP 使用了线性规划模型(LP)求解

Objective value:136.2275 目标函数最小值为136.2275

Infeasibilities:0 所有约束条件均得到满足

运输方案

工地	A料场供给(t)	B料场供给(t)	总需求(t)
1	3.0	0.0	3.0
2	5.0	0.0	5.0
3	0.0	4.0	4.0
4	7.0	0.0	7.0
5	0.0	6.0	6.0
6	1.0	10.0	11.0
总计	16.0	20.0	36.0

Part2: 选址优化

模型变量

1.决策变量

- xj_j 、 yj_j : 料场的横纵坐标
- x_{ij} : 从料场 j 向工地 i 运送的水泥量

2.其他变量

- xi_i 、 yi_i : 工地的横纵坐标
 - d_{ij} : 料场 j 到工地 i 的距离
 - a_i : 工地 i 的日用量
 - s_j : 料场 j 的最大日运输量
- 其中 $i = 1, 2, 3, 4, 5, 6; j = 1, 2$ (对应A, B)

目标函数

最小化：

$$\sum_{i=1}^6 \sum_{j=1}^2 x_{ij} d_{ij}$$

约束条件不变

...

代码实现

```
SETS:
site /1..6/: xi, yi, ai;
cement /1..2/: supply, xj, yj;
transport(site,cement): x;
ENDSETS

DATA:
xi = 1.25 8.75 0.5 5.75 3 7.25;
yi = 1.25 0.75 4.75 5 6.5 7.75;
ai = 3 5 4 7 6 11;
supply = 20 20;
ENDDATA

! 目标函数：总吨·公里数；
min = @sum(transport(i,j): x(i,j) * @sqrt((xi(i) - xj(j))^2 + (yi(i) - yj(j))^2));

! 工地需求约束；
@for(site(i):
    @sum(cement(j): x(i,j)) = ai(i)
);

! 料场供应限制；
@for(cement(j):
    @sum(site(i): x(i,j)) <= supply(j)
);

END
```

运行结果解读

Model Class:NLP 使用了非线性规划模型(NLP)求解

Objective value:85.26604 目标函数最小值为85.26604，相较于第一问运输成本下降了约37%

Infeasibilities:0 所有约束条件均得到满足

最优料场位置

料场	X坐标	Y坐标
A	3.2549	5.6523
B	7.2500	7.7500

运输方案

工地	A料场供给(t)	B料场供给(t)	总需求(t)
1	3.0	0.0	3.0
2	0.0	5.0	5.0
3	4.0	0.0	4.0
4	7.0	0.0	7.0
5	6.0	0.0	6.0
6	0.0	11.0	11.0
合计	20.0	16.0	36.0
